
N16R8
Der große Bruder des ESP32:

Vektorbefehle für KI, 8 MB PSRAM und

nativer USB. Erster Upload in rund

20 Minuten, danach sechs Beispiele – bis

zu Sprachbefehlen, die ganz ohne Internet

01 Auf einen Blick
Das Board trägt ein ESP32-S3-WROOM-1 in der größten Speicherausstattung N16R8: 16 MB

Flash und 8 MB PSRAM. Programmiert wird es wie jeder ESP32 mit der Arduino IDE – nur ein

paar Einstellungen sind anders.

Technische Daten

Modul ESP32-S3-WROOM-1-

N16R8

Prozessor 2 × Xtensa LX7, bis 240

MHz, mit Vektorbefehlen

Speicher 16 MB Flash (Quad) · 8 MB

PSRAM (Octal) · 512 kB

SRAM

Funk WLAN 2,4 GHz (b/g/n) ·

Bluetooth LE 5.0

USB 2 × USB-C: COM (USB-

UART) und USB (nativ,

OTG)

Versorgung 5 V über USB-C oder 5Vin,

Regler AMS1117-3.3

Logikpegel 3,3 V – Pins nicht 5-V-

tolerant

Anschlüsse 2 × 22 Pins, 2,54-mm-

Raster (YD-ESP32-S3-

Layout)

Lieferumfang

1 × ESP32-S3 N16R8 Dev Kit

1 × Stiftleisten-Set (zum Einlöten)

! 3,3 V an den Pins

Die GPIOs vertragen keine 5 V. 5-V-

Signale nur über Spannungsteiler oder

Pegelwandler anschließen.

i Stiftleisten einlöten

Die beiden 22-poligen Leisten liegen bei.

Mit der langen Seite nach unten ins

Steckbrett stecken, Board auflegen, alle

Pins verlöten.

ESP32 oder ESP32-S3?

ESP32 DEVKIT ESP32-S3 N16R8 (DIESES BOARD)

Prozessor 2 × LX6, 240 MHz 2 × LX7, 240 MHz, Vektorbefehle für KI

Speicher 4 MB Flash, 520 kB SRAM 16 MB Flash, 8 MB PSRAM, 512 kB SRAM

Bluetooth Classic + LE nur LE 5.0

USB nur über Wandler Wandler und nativer USB (Tastatur, MIDI …)

Was du zusätzlich brauchst

USB-C-Datenkabel

Kein reines Ladekabel – es muss Daten

übertragen

 

PC mit Arduino IDE 2

Windows, macOS oder Linux, kostenlos

 

Für die KI-Sprachsteuerung

INMP441 I2S-Mikrofon + Steckbrett,

Jumper

MKR-INMP441-64

MAKEROO · support@makeroo.de ESP32-S3 N16R8 · MKR-ESP32-S3-008

02 Dein Board im Detail
Das Board folgt dem verbreiteten YD-ESP32-S3-Layout: zwei USB-C-Buchsen, RGB-LED,

zwei Tasten und 2 × 22 Pins. Die Nummern im Foto gehören zur Tabelle.

1 ESP32-S3-WROOM-1 – Das Funkmodul mit

Prozessor, 16 MB Flash und 8 MB PSRAM unter dem

Blechdeckel. Die Variante steht auf dem Etikett bzw.

Karton: N16R8.

2 PCB-Antenne – WLAN und Bluetooth. Kein Metall

darüber oder darunter, nicht auf leitende Flächen

legen.

3 USB-C-Buchse – Eine der beiden Buchsen heißt

COM – über den USB-UART-Wandler (5) zum

seriellen Monitor. Damit lädst du hoch.

4 USB-C-Buchse – Die andere heißt USB – direkt am

nativen USB des S3 (GPIO19/20). Für USB-Tastatur,

MIDI, Massenspeicher.

5 USB-UART-Wandler – Macht aus „COM“ einen

seriellen Port und schaltet beim Hochladen

automatisch in den Boot-Modus.

6 RGB-LED – Adressierbare WS2812 an GPIO48.

Ansteuerung mit rgbLedWrite(48, r, g, b).

7 Taste BOOT – Hängt an GPIO0. Gedrückt beim

Einschalten: Download-Modus. Im Betrieb als Taster

nutzbar.

8 Taste RST – Neustart des Sketches.

9 Spannungsregler – AMS1117 macht aus 5 V die 3,3 V

für Chip und den 3V3-Pin.

10 Status-LEDs – Power sowie die Datenleitungen

TX/RX des COM-Ports.

11 Lötbrücke IN-OUT – Offen ist 5Vin nur ein Eingang.

Geschlossen liegt die USB-Spannung auch an 5Vin

an – nur schließen, wenn du 5 V abgreifen willst.

i Welche Buchse wofür?

Hochladen und serieller Monitor: immer COM. Die Buchse USB brauchst du erst, wenn der S3 selbst

ein USB-Gerät sein soll (Beispiel 3). Beide Buchsen versorgen das Board – ein Kabel genügt.

! Drei Pins sind tabu

GPIO35, GPIO36 und GPIO37 stehen auf der Leiste, sind beim N16R8 aber intern mit dem Octal-

PSRAM verbunden. Dort nichts anschließen.

MAKEROO · support@makeroo.de ESP32-S3 N16R8 · MKR-ESP32-S3-008

03 Pinbelegung
Antenne oben, USB-Buchsen unten. Auf der Platine steht nur die Nummer – im Code schreibst

du genau diese Zahl, z. B. digitalWrite(21, HIGH).

Versorgung Masse frei nutzbar ADC1 · Touch I2C SPI UART0 / COM nativer USB

Strapping – Vorsicht nicht nutzbar RGB-LED

ESP32-S3

WROOM-1

N16R8

AMS1117

RGB

RST

BOOT

UART

USBC USBC

3V3

3V3

RST

4

5

6

7

15

16

17

18

8

3

46

9

10

11

12

13

14

5Vin

GND

GND

TX

RX

1

2

42

41

40

39

38

37

36

35

0

45

48

47

21

20

19

GND

GND

3,3-V-Ausgang 3V3

3,3-V-Ausgang 3V3

Reset (aktiv LOW) RST

ADC1 · Touch GPIO4 4

ADC1 · Touch GPIO5 5

ADC1 · Touch GPIO6 6

ADC1 · Touch GPIO7 7

frei · ADC2 GPIO15 15

frei · ADC2 GPIO16 16

frei · ADC2 GPIO17 17

frei · ADC2 GPIO18 18

I2C SDA · ADC1 GPIO8 8

Strapping (JTAG) GPIO3 3

Strapping GPIO46 46

I2C SCL · ADC1 GPIO9 9

SPI CS · ADC1 GPIO10 10

SPI MOSI GPIO11 11

SPI SCK GPIO12 12

SPI MISO GPIO13 13

frei · ADC2 · Touch GPIO14 14

5 V Ein-/Ausgang 5Vin

Masse GND

GND Masse

TX GPIO43 UART0 → COM-Port

RX GPIO44 UART0 ← COM-Port

1 GPIO1 ADC1 · Touch

2 GPIO2 ADC1 · Touch

42 GPIO42 frei (JTAG)

41 GPIO41 frei (JTAG)

40 GPIO40 frei (JTAG)

39 GPIO39 frei (JTAG)

38 GPIO38 frei

37 GPIO37 belegt – PSRAM

36 GPIO36 belegt – PSRAM

35 GPIO35 belegt – PSRAM

0 GPIO0 BOOT-Taste · Strapping

45 GPIO45 Strapping

48 GPIO48 RGB-LED (WS2812)

47 GPIO47 frei

21 GPIO21 frei

20 GPIO20 nativer USB D+

19 GPIO19 nativer USB D−

GND Masse

GND Masse

MAKEROO · support@makeroo.de ESP32-S3 N16R8 · MKR-ESP32-S3-008

04 Einrichten
Die Schritte 1 und 2 sind dieselben wie beim ESP32. Entscheidend beim S3 sind die

Einstellungen im Menü Werkzeuge – vor allem PSRAM.

1 Arduino IDE installieren

Aktuelle Arduino IDE 2 von arduino.cc/en/software laden und installieren.

2 ESP32-Board-Paket installieren

Links auf das Board-Symbol klicken, esp32 eingeben und beim Eintrag

esp32 von Espressif Systems auf Installieren klicken (rund 300 MB).

Version 3.0 oder neuer ist Pflicht – die KI-Beispiele brauchen sie. Nicht

„Arduino ESP32 Boards“ nehmen.

Das Bild stammt aus unserem ESP32-Guide; der Eintrag ist für ESP32 und

ESP32-S3 derselbe.

3 Board anschließen, Board und Port wählen

USB-C-Kabel an die Buchse COM. In der Board-Auswahl ESP32S3 Dev Module suchen und den neuen

COM-Port wählen. Kein Port? Siehe Seite 15.

4 Einstellungen unter Werkzeuge

Einmal setzen, die IDE merkt sie sich. Nicht genannte Punkte bleiben auf dem Standardwert.

MENÜPUNKT WERT WARUM

USB CDC On Boot Disabled Serielle Ausgabe läuft über die COM-Buchse.

Flash Mode QIO 80MHz Der Flash des N16R8 ist Quad-SPI.

Flash Size 16MB (128Mb) Sonst nutzt der Sketch nur 4 MB.

Partition Scheme 16M Flash (3MB APP/9.9MB FATFS) 3 MB für den Sketch. Für Beispiel 6: „ESP SR 16M

…“.

PSRAM OPI PSRAM Wichtig! Das „R8“ ist Octal-PSRAM. Falsch

gewählt: 0 MB PSRAM oder Neustart-Schleife.

USB Mode Hardware CDC and JTAG Standard. Für Beispiel 3: „USB-OTG (TinyUSB)“.

MAKEROO · support@makeroo.de ESP32-S3 N16R8 · MKR-ESP32-S3-008

05 Hochladen & Chip prüfen
Der erste Sketch liest aus, was wirklich auf deinem Board steckt – Chip, Flash und PSRAM –

und lässt die RGB-LED durch die Farben laufen.

Kompiliert für „ESP32S3 Dev Module“: 298 kB Flash (9 %), 22 kB RAM (6 %).

Zeile 4

RGB-LED

Die WS2812-LED „RGB“ hängt an

GPIO48.

Zeile 10

Echte Flash-Größe

Liest die Größe aus dem Chip –

unabhängig von der Einstellung in

der IDE.

Zeile 15–16

PSRAM prüfen

8 MB erwartet. 0 MB heißt: PSRAM-

Einstellung falsch.

SO SIEHT DIE AUSGABE AUS · 115200 BAUD

MAKEROO ESP32-S3 Dev Kit
Chip:   ESP32-S3, Revision x, 2 Kerne, 240 MHz

Flash:  16 MB
PSRAM:  8 MB (frei: ~8 100 kB)

RAM:    ~3xx kB frei im internen Speicher
RGB-LED: rot
RGB-LED: gruen

Hochladen

Pfeil-Symbol oder Strg  + U . Das erste Kompilieren

dauert etwa eine Minute. Danach den Monitor öffnen

( Strg  + Umschalt  + M ) und 115200 Baud einstellen;

ggf. RST drücken.

Was die Zahlen bedeuten

16 MB und 8 MB bestätigen ein N16R8. Revision und freier

RAM schwanken je nach Charge und Board-Paket.

MAKEROO_ESP32S3_Start.ino

// MAKEROO · ESP32-S3 N16R8 · Erster Start: Chip pruefen und RGB-LED1
#include "esp_flash.h"2
 3
const int RGB_PIN = 48;                          // WS2812 auf dem Board ("RGB")4
 5
void setup() {6
  Serial.begin(115200);7
  delay(1000);8
  uint32_t flash = 0;9
  esp_flash_get_physical_size(NULL, &flash);     // echte Chipgroesse10
  Serial.println("\nMAKEROO ESP32-S3 Dev Kit");11
  Serial.printf("Chip:   %s, Revision %d, %d Kerne, %d MHz\n", ESP.getChipModel(),12
                ESP.getChipRevision(), ESP.getChipCores(), (int)ESP.getCpuFreqMHz());13
  Serial.printf("Flash:  %d MB\n", (int)(flash / (1024 * 1024)));14
  Serial.printf("PSRAM:  %d MB (frei: %d kB)\n", (int)(ESP.getPsramSize() / (1024 * 1024)),15
                (int)(ESP.getFreePsram() / 1024));16
  Serial.printf("RAM:    %d kB frei im internen Speicher\n", (int)(ESP.getFreeHeap() / 1024));17
  if (ESP.getPsramSize() == 0) {18
    Serial.println("Kein PSRAM! Werkzeuge -> PSRAM -> \"OPI PSRAM\" waehlen.");19
  }20
}21
 22
void loop() {23
  const uint8_t farbe[5][3] = {{40, 0, 0}, {0, 40, 0}, {0, 0, 40}, {25, 25, 25}, {0, 0, 0}};24
  const char* name[5] = {"rot", "gruen", "blau", "weiss", "aus"};25
  for (int i = 0; i < 5; i++) {26
    rgbLedWrite(RGB_PIN, farbe[i][0], farbe[i][1], farbe[i][2]);27
    Serial.printf("RGB-LED: %s\n", name[i]);28
    delay(800);29
  }30
}31

MAKEROO · support@makeroo.de ESP32-S3 N16R8 · MKR-ESP32-S3-008

06 Beispiele: Was der S3 kann
Beispiel 1 8 MB PSRAM nutzen KEIN ZUBEHÖR

Der interne Arbeitsspeicher hat 512 kB. Mit ps_malloc() holst du dir Speicher aus dem 8-MB-PSRAM – hier einen 4-

MB-Puffer, wie ihn Kamerabilder, Audio oder KI-Modelle brauchen.

Kompiliert für „ESP32S3 Dev Module“: 273 kB Flash (8 %), 22 kB RAM (6 %).

BEISPIELAUSGABE · 115200 BAUD

PSRAM gesamt: ~8 190 kB, frei: ~8 100 kB

4 MB geschrieben in … ms, gelesen in … ms
Pruefsumme: 534773760 (erwartet 534773760)
PSRAM frei mit Puffer: ~4 000 kB

Gut zu wissen

Stimmt die Prüfsumme, wurde jedes Byte richtig

gespeichert. PSRAM ist langsamer als der interne RAM –

zeitkritische Daten bleiben besser dort.

Wofür braucht man so viel Speicher?

Kamerabild

Ein Bild mit 320 × 240 Pixeln in 16-

Bit-Farbe belegt 150 kB – mehr als ein

Viertel des internen RAM. Im PSRAM

passen über 50 davon.

Audio

10 Sekunden Sprache mit 16 kHz und

16 Bit sind 313 kB. Spracherkennung

(Beispiel 6) puffert ihr Audio genau

dort.

Display

Ein Bildspeicher für ein 480 × 320-

Display braucht 300 kB – damit

zeichnest du flimmerfrei im

Hintergrund.

i So landen Daten im PSRAM

ps_malloc(größe) oder heap_caps_malloc(größe, MALLOC_CAP_SPIRAM) statt malloc().

Große Puffer nicht als globale Arrays anlegen – die landen im internen RAM. Viele Bibliotheken

(Kamera, TensorFlow Lite, LVGL) nutzen PSRAM automatisch, wenn er aktiviert ist.

S3_01_PSRAM.ino

// MAKEROO · ESP32-S3 · Beispiel 1: 8 MB PSRAM nutzen1
void setup() {2
  Serial.begin(115200);3
  delay(1000);4
  Serial.printf("PSRAM gesamt: %d kB, frei: %d kB\n",5
                (int)(ESP.getPsramSize() / 1024), (int)(ESP.getFreePsram() / 1024));6
  const size_t GROESSE = 4 * 1024 * 1024;       // 4 MB - zu gross fuer den internen RAM7
  uint8_t* puffer = (uint8_t*)ps_malloc(GROESSE);8
  if (puffer == nullptr) {9
    Serial.println("ps_malloc fehlgeschlagen - PSRAM aktiviert?");10
    return;11
  }12
  uint32_t t0 = millis();13
  for (size_t i = 0; i < GROESSE; i++) puffer[i] = i & 0xFF;   // schreiben14
  uint32_t t1 = millis();15
  uint32_t summe = 0;16
  for (size_t i = 0; i < GROESSE; i++) summe += puffer[i];    // lesen17
  uint32_t t2 = millis();18
  Serial.printf("4 MB geschrieben in %d ms, gelesen in %d ms\n", (int)(t1 - t0), (int)(t2 - t1));19
  Serial.printf("Pruefsumme: %u (erwartet 534773760)\n", (unsigned)summe);20
  Serial.printf("PSRAM frei mit Puffer: %d kB\n", (int)(ESP.getFreePsram() / 1024));21
  free(puffer);22
}23
 24
void loop() {}25

MAKEROO · support@makeroo.de ESP32-S3 N16R8 · MKR-ESP32-S3-008

Beispiel 2 Berührung erkennen DRAHT ODER ALUFOLIE

GPIO1 bis GPIO14 sind Touch-Eingänge. Ein Draht an GPIO4 genügt – bei Berührung leuchtet die RGB-LED grün.

Kompiliert für „ESP32S3 Dev Module“: 314 kB Flash (10 %), 22 kB RAM (7 %).

Ergebnis

Im Monitor läuft der Messwert mit,

zuerst der Ruhewert. Berührst du den

Draht, steigt die Zahl deutlich,

dahinter steht „<- beruehrt“ und die

RGB-LED leuchtet grün.

Anders als beim ESP32

Beim S3 steigt der Messwert bei

Berührung, beim alten ESP32 sinkt

er. Der Sketch merkt sich beim Start

den Ruhewert – dabei den Draht

nicht berühren.

Beispiel 3 Das Board als USB-Tastatur ZWEITES USB-C-KABEL

Der S3 hat USB direkt im Chip. Damit meldet er sich am PC als Tastatur – ein Druck auf BOOT tippt einen Text. So

baust du Makro-Tasten, Fußschalter oder Stream-Decks.

Kompiliert für „ESP32S3 Dev Module“ · USB Mode: USB-OTG (TinyUSB): 357 kB Flash (11 %), 54 kB RAM (16 %).

So testest du

1.

Werkzeuge → USB Mode → USB-OTG (TinyUSB).
2. Über COM hochladen.

3. Zweites Kabel an USB – der PC meldet eine neue

Tastatur.

4. Editor öffnen, BOOT drücken.

Achtung

Die Tastatur tippt in jedes

Fenster, das gerade aktiv

ist. Erst einen leeren Editor

öffnen.

Mehr

Im Board-Paket stecken weitere

USB-Beispiele: Maus, Gamepad,

MIDI und USB-Stick

(Datei → Beispiele → USB).

S3_02_Touch.ino

// MAKEROO · ESP32-S3 · Beispiel 2: Beruehrung erkennen1
const int TOUCH_PIN = 4;          // Touch T4: Draht oder Alufolie2
const int RGB_PIN = 48;3
uint32_t ruhe;4
 5
void setup() {6
  Serial.begin(115200);7
  delay(1000);8
  ruhe = touchRead(TOUCH_PIN);      // beim Start nicht beruehren!9
  Serial.printf("Ruhewert: %u\n", (unsigned)ruhe);10
}11
 12
void loop() {13
  uint32_t wert = touchRead(TOUCH_PIN);14
  bool beruehrt = wert > ruhe * 1.2; // beim S3 STEIGT der Wert15
  rgbLedWrite(RGB_PIN, 0, beruehrt ? 40 : 0, 0);16
  Serial.printf("%u%s\n", (unsigned)wert, beruehrt ? "  <- beruehrt" : "");17
  delay(200);18
}19

S3_03_USB_Tastatur.ino

// MAKEROO · ESP32-S3 · Beispiel 3: Das Board als USB-Tastatur1
// Werkzeuge -> USB Mode: "USB-OTG (TinyUSB)". Danach Kabel an die Buchse "USB".2
#include "USB.h"3
#include "USBHIDKeyboard.h"4
 5
USBHIDKeyboard tastatur;6
const int TASTE = 0;                             // die BOOT-Taste7
 8
void setup() {9
  pinMode(TASTE, INPUT_PULLUP);10
  tastatur.begin();11
  USB.begin();                                   // meldet sich am PC als Tastatur12
}13
 14
void loop() {15
  if (digitalRead(TASTE) == LOW) {16
    tastatur.print("Hallo von MAKEROO! ");17
    delay(300);18
    while (digitalRead(TASTE) == LOW) delay(10); // warten bis losgelassen19
  }20
}21

MAKEROO · support@makeroo.de ESP32-S3 N16R8 · MKR-ESP32-S3-008

Beispiel 4 Farbe per Webseite SMARTPHONE

Der S3 öffnet sein eigenes WLAN „MAKEROO-S3“ und liefert eine Seite mit Farbwähler. Die RGB-LED übernimmt

sofort jede gewählte Farbe.

Kompiliert für „ESP32S3 Dev Module“: 889 kB Flash (28 %), 46 kB RAM (14 %).

192.168.4.1

Seite aus dem Sketch

So testest du

1. Hochladen.

2. Am Handy WLAN MAKEROO-S3,

Passwort makeroo123.

3. Im Browser 192.168.4.1 öffnen.

4. Auf die Farbfläche tippen und eine Farbe

wählen.

BEISPIELAUSGABE · 115200 BAUD

Hotspot aktiv: http://192.168.4.1

Gut zu wissen

Die Farbwerte werden durch 4 geteilt –

volle Helligkeit blendet. Passwort vor dem

Dauerbetrieb ändern.

S3_04_Webserver.ino

// MAKEROO · ESP32-S3 · Beispiel 4: RGB-LED per Webseite1
#include <WiFi.h>2
#include <WebServer.h>3
 4
const int RGB_PIN = 48;5
WebServer server(80);6
 7
const char SEITE[] = R"(<!doctype html><meta name=viewport content="width=device-width">8
<body style="font-family:sans-serif;text-align:center"><h2>MAKEROO ESP32-S3</h2>9
<p>Farbe waehlen:</p><input type=color id=c value="#00c0ff" style="width:80%;height:50vh">10
<script>c.oninput=()=>fetch('/farbe?c='+c.value.slice(1))</script>)";11
 12
void setup() {13
  Serial.begin(115200);14
  WiFi.softAP("MAKEROO-S3", "makeroo123");      // eigenes WLAN15
  Serial.print("Hotspot aktiv: http://");16
  Serial.println(WiFi.softAPIP());17
  server.on("/", [] { server.send(200, "text/html", SEITE); });18
  server.on("/farbe", [] {19
    long c = strtol(server.arg("c").c_str(), nullptr, 16);   // "ff8800" -> Zahl20
    rgbLedWrite(RGB_PIN, (c >> 16) / 4, ((c >> 8) & 0xFF) / 4, (c & 0xFF) / 4);21
    server.send(200, "text/plain", "ok");22
  });23
  server.begin();24
}25
 26
void loop() {27
  server.handleClient();28
}29

07 KI auf dem ESP32-S3 – was geht wirklich?
KI auf einem Mikrocontroller heißt: Ein kleines, fertig trainiertes neuronales Netz läuft direkt

auf dem Chip – ohne Cloud, ohne Internet, in Millisekunden. Trainiert wird am PC, der S3 führt

das Modell aus. Fachleute nennen das TinyML oder Edge-KI.

Vektorbefehle

Die LX7-Kerne verarbeiten mehrere 8-Bit-Werte in einem

Befehl – genau die Rechnung, aus der neuronale Netze

bestehen. Espressifs Bibliothek ESP-NN nutzt das. Laut

Espressif sinkt die Rechenzeit eines Personen-

Erkennungsmodells damit von 2,3 s auf 54 ms (240 MHz).

Speicher

8 MB PSRAM für Modelle, Audio- und Bildpuffer, 16 MB

Flash für die Modelldateien. Allein das Sprachmodell aus

Beispiel 6 belegt rund 3,3 MB.

Zwei Kerne

Ein Kern rechnet die KI, der andere kümmert sich um WLAN,

Display und Sensoren.

Ehrlich gesagt

ChatGPT & Co. laufen nicht auf dem Chip – solche Modelle

sind tausendfach größer. Der S3 kann sie aber per WLAN

als Cloud-Dienst nutzen.

Was geht mit diesem Board?

AUFGABE MACHBAR? WERKZEUG

Aktivierungswort + Sprachbefehle,

offline

Ja – mit INMP441 ESP-SR im Board-Paket (Beispiel 6), ESPHome

micro_wake_word

Kleine eigene Netze Ja TensorFlow Lite Micro (Beispiel 5)

Sensordaten deuten: Gesten,

Vibration, Anomalien

Ja – mit Sensor, z. B.

MPU-6050

Edge Impulse, TensorFlow Lite Micro

Geräusche erkennen (Klatschen,

Glas, Hund)

Ja – mit INMP441 Edge Impulse

Bilder: Personen, Gesichter, Objekte Nur mit Kamera – dieses

Board hat keinen

Kameraanschluss

ESP-WHO, ESP-DL auf einem S3-Kameraboard

Große Sprachmodelle (ChatGPT,

Claude …)

Nicht lokal – nur per

WLAN-API

HTTPS-Anfrage an einen Cloud-Dienst

Modelle trainieren Nein – am PC oder in der

Cloud

Python/TensorFlow, Edge Impulse

So kommt ein Modell auf den Chip

1

Daten sammeln

Sensorwerte oder

Audio aufnehmen –

oft mit dem S3 selbst.

2

Trainieren

Am PC oder im

Browser (Edge

Impulse).

3

Verkleinern

Quantisieren: 32-Bit-

Kommazahlen

werden 8-Bit-

Ganzzahlen.

4

Exportieren

Als C-Array

(model.h) oder

Arduino-Bibliothek.

5

Ausführen

Der S3 rechnet das

Netz – die „Inferenz“.

KI-Beispiel 5 Ein neuronales Netz rechnet sin(x) KEIN ZUBEHÖR

Das „Hallo Welt“ von TensorFlow Lite Micro: Ein winziges Netz (3 Schichten, 2,4 kB) hat gelernt, die Sinuskurve

nachzubilden. Der S3 rechnet es für 13 Werte und vergleicht mit dem echten Sinus. Die RGB-LED pulsiert im Takt der

Kurve.

Kompiliert für „ESP32S3 Dev Module“ · mit Chirale_TensorFlowLite 2.0.0: 577 kB Flash (18 %), 31 kB RAM (9 %).

Ergebnis des Netzes

X KI ECHT

0.0 +0.00 +0.00

0.5 +0.48 +0.48

1.0 +0.88 +0.84

1.5 +1.04 +1.00

2.0 +0.88 +0.91

2.5 +0.61 +0.60

3.0 +0.13 +0.14

3.5 −0.31 −0.35

4.0 −0.79 −0.76

4.5 −1.02 −0.98

5.0 −0.99 −0.96

5.5 −0.66 −0.71

6.0 −0.33 −0.28

Vorbereitung

1. Bibliotheksverwalter:

Chirale_TensorFlowLite installieren.

2. Sketch-Ordner aus dem Download

öffnen – model.h muss daneben

liegen.

3. Hochladen, Monitor auf 115200

Baud.

Was du siehst

Jede Zeile zeigt x, die KI-Antwort, den

echten Sinus und die Rechenzeit in

Mikrosekunden. Die KI-Werte in der

Tabelle haben wir mit demselben

Modell am PC nachgerechnet.

Quantisierung

Das Netz rechnet mit 8-Bit-Ganzzahlen

statt Kommazahlen – viermal kleiner

und schneller. Dafür liegt es um ein

paar Hundertstel daneben. Die beiden

Zeilen mit params.scale rechnen hin

und zurück.

Schneller geht es

Diese Bibliothek nutzt allgemeine

Rechenfunktionen. Mit Espressifs esp-

tflite-micro (ESP-IDF) kommen die

Vektorbefehle über ESP-NN dazu – bei

größeren Modellen ein Vielfaches

schneller.

S3_05_KI_Sinus.ino

// MAKEROO · ESP32-S3 · KI-Beispiel 5: Ein neuronales Netz rechnet sin(x)1
// Bibliothek "Chirale_TensorFlowLite" installieren, model.h liegt im Sketch-Ordner2
#include <Chirale_TensorFlowLite.h>3
#include "model.h"4
#include "tensorflow/lite/micro/all_ops_resolver.h"5
#include "tensorflow/lite/micro/micro_interpreter.h"6
#include "tensorflow/lite/schema/schema_generated.h"7
 8
const int RGB_PIN = 48;9
alignas(16) uint8_t arena[2048];                 // Arbeitsspeicher des Netzes10
tflite::MicroInterpreter* netz;11
TfLiteTensor *ein, *aus;12
 13
void setup() {14
  Serial.begin(115200);15
  static tflite::AllOpsResolver ops;16
  static tflite::MicroInterpreter interpreter(tflite::GetModel(g_model), ops, arena, sizeof(arena));17
  netz = &interpreter;18
  if (netz->AllocateTensors() != kTfLiteOk) {19
    Serial.println("Arena zu klein!");20
    while (true) delay(10);21
  }22
  ein = netz->input(0);23
  aus = netz->output(0);24
}25
 26
void loop() {27
  for (float x = 0; x < 6.3; x += 0.5) {28
    ein->data.int8[0] = x / ein->params.scale + ein->params.zero_point;   // quantisieren29
    uint32_t t = micros();30
    netz->Invoke();                                                       // Inferenz31
    t = micros() - t;32
    float y = (aus->data.int8[0] - aus->params.zero_point) * aus->params.scale;33
    Serial.printf("x = %.1f   KI: %+.2f   echt: %+.2f   (%d us)\n", x, y, sin(x), (int)t);34
    rgbLedWrite(RGB_PIN, 0, 0, (uint8_t)((y + 1) * 30));                 // Helligkeit folgt35
    delay(300);36
  }37
}38

KI-Beispiel 6 Sprachbefehle ohne Internet INMP441-MIKROFON

Mit dem Spracherkenner ESP-SR, der im Board-Paket steckt, hört der S3 auf das Aktivierungswort „Hi ESP“ und

versteht danach englische Befehle. Zwei neuronale Netze laufen dabei auf dem Chip: WakeNet für das

Aktivierungswort, MultiNet für die Befehle. Keine Cloud, kein Konto.

INMP441 ESP32-S3

VDD → 3V3

GND → GND

L/R → GND

SCK → 17

WS → 15

SD → 16

! Zwei Einstellungen sind Pflicht

Partition Scheme → ESP SR 16M (3MB APP/6MB SPIFFS/3.9MB MODEL)

– sonst fehlt das Sprachmodell. Und PSRAM → OPI PSRAM. Die IDE

lädt das 3,3-MB-Modell beim Hochladen automatisch mit.

i L/R an GND

Damit liefert das Mikrofon den linken Kanal – genau den liest der

Sketch. Kabel kurz halten (unter 15 cm).

STECKPLATINEN-ANSICHT · FRITZING

Kompiliert für „ESP32S3 Dev Module“ · Partition Scheme: ESP SR 16M: 2393 kB Flash (77 %), 40 kB RAM (12 %).

Befehle im Sketch

Hi ESP aufwecken

Turn on the
light

Licht an

(weiß)

Turn off the
light

Licht aus

Set color red rot

Set color green grün

Set color blue blau

BEISPIELAUSGABE · 115200 BAUD

Sag "Hi ESP" ...

Hi ESP erkannt - sag einen Befehl
Befehl: Licht an
Befehl: Blau

Zeit abgelaufen -
 warte auf "Hi ESP"

So klappt es

Deutlich, mit normaler Lautstärke aus

0,5 bis 2 m sprechen. „Hi ESP“

englisch: „Hai I-Es-Pi“. Ruhige

Umgebung hilft.

Eigene Befehle

Einfach Zeilen in die Liste befehle[]

eintragen – auf Englisch, mit eigener

Nummer. MultiNet erkennt sie ohne

neues Training. Deutsche Befehle

unterstützt dieses Modell nicht.

Mehr Sprache

Ein Aktivierungswort wie „Okay

Nabu“ und Befehle in Deutsch

bekommst du mit ESPHome + Home

Assistant (Seite 14).

S3_06_KI_Sprache.ino

// MAKEROO · ESP32-S3 · KI-Beispiel 6: Sprachbefehle offline (INMP441)1
// Werkzeuge -> Partition Scheme: "ESP SR 16M ...", PSRAM: "OPI PSRAM"2
#include "ESP_I2S.h"3
#include "ESP_SR.h"4
 5
const int RGB_PIN = 48;6
const int MIC_SCK = 17, MIC_WS = 15, MIC_SD = 16;   // INMP441, L/R an GND7
I2SClass i2s;8
 9
enum { AN, AUS, ROT, GRUEN, BLAU };10
const char* NAME[] = {"Licht an", "Licht aus", "Rot", "Gruen", "Blau"};11
static const sr_cmd_t befehle[] = {12
  {AN, "Turn on the light"}, {AN, "Lights on"}, {AUS, "Turn off the light"},13
  {AUS, "Lights off"}, {ROT, "Set color red"}, {GRUEN, "Set color green"},14
  {BLAU, "Set color blue"},15
};16
 17
void ereignis(sr_event_t e, int befehl, int phrase) {18
  if (e == SR_EVENT_WAKEWORD) {                 // "Hi ESP" gehoert19
    Serial.println("Hi ESP erkannt - sag einen Befehl");20
    ESP_SR.setMode(SR_MODE_COMMAND);21
  } else if (e == SR_EVENT_TIMEOUT) {           // Befehl kam nicht rechtzeitig22
    Serial.println("Zeit abgelaufen - warte auf \"Hi ESP\"");23
    ESP_SR.setMode(SR_MODE_WAKEWORD);24
  } else if (e == SR_EVENT_COMMAND) {25
    Serial.printf("Befehl: %s\n", NAME[befehl]);26
    const uint8_t f[5][3] = {{30, 30, 30}, {0, 0, 0}, {40, 0, 0}, {0, 40, 0}, {0, 0, 40}};27
    rgbLedWrite(RGB_PIN, f[befehl][0], f[befehl][1], f[befehl][2]);28
    ESP_SR.setMode(SR_MODE_COMMAND);            // weitere Befehle erlaubt29
  }30
}31
 32
void setup() {33
  Serial.begin(115200);34
  i2s.setPins(MIC_SCK, MIC_WS, -1, MIC_SD);35
  i2s.setTimeout(1000);36
  i2s.begin(I2S_MODE_STD, 16000, I2S_DATA_BIT_WIDTH_32BIT, I2S_SLOT_MODE_MONO, I2S_STD_SLOT_LEFT);37
  i2s.configureRX(16000, I2S_DATA_BIT_WIDTH_32BIT, I2S_SLOT_MODE_MONO, I2S_RX_TRANSFORM_32_TO_16);38
  ESP_SR.onEvent(ereignis);39
  ESP_SR.begin(i2s, befehle, sizeof(befehle) / sizeof(sr_cmd_t), SR_CHANNELS_MONO, SR_MODE_WAKEWORD, "M");40
  Serial.println("Sag \"Hi ESP\" ...");41
}42
 43
void loop() {44
  delay(100);                                   // alles laeuft im Hintergrund45
}46

08 KI: Wie geht's weiter?
Eigene Modelle mit Edge Impulse

Die bequemste Art, eigene KI zu trainieren – im Browser, ohne

Programmierkenntnisse in Python.

1 Projekt anlegen

Kostenlos auf edgeimpulse.com.

2 Daten aufnehmen

Z. B. Bewegungen mit dem MPU-6050 oder Geräusche mit dem

INMP441 – je Klasse ein paar Minuten.

3 Trainieren

Im Browser ein Modell wählen und trainieren lassen. Edge

Impulse zeigt Genauigkeit und Speicherbedarf.

4 Als Arduino-Bibliothek exportieren

Die ZIP-Datei über

Sketch → Bibliothek einbinden → .ZIP-Bibliothek hinzufügen

installieren und das mitgelieferte ESP32-Beispiel öffnen.

Sprachassistent mit ESPHome

Mit dem INMP441 wird der S3 zum

lokalen Sprachassistenten für Home

Assistant: Das Aktivierungswort erkennt

micro_wake_word auf dem Chip, die

Sprache verarbeitet dein Home-

Assistant-Server. Grundgerüst für dieses

Board:

esp32:

  board: esp32-s3-devkitc-1

  flash_size: 16MB

  framework:

    type: esp-idf

psram:

  mode: octal

  speed: 80MHz

light:

  - platform: esp32_rmt_led_strip

    name: "RGB-LED"

    pin: GPIO48

    num_leds: 1

    rgb_order: GRB

    chipset: ws2812

Dazu kommen i2s_audio, microphone (Pins

wie in Beispiel 6), micro_wake_word und

voice_assistant – Anleitung auf

esphome.io.

Weitere Wege

ESP-IDF + ESP-DL

Espressifs eigene KI-Bibliothek mit

fertigen Modellen und voller Nutzung

der Vektorbefehle. Für

Fortgeschrittene, die von der Arduino

IDE auf ESP-IDF umsteigen.

Kamera-KI

Personen- und Gesichtserkennung

(ESP-WHO) brauchen eine Kamera –

dafür gibt es eigene ESP32-S3-

Kameraboards.

Cloud-KI

Der S3 kann per HTTPS einen Cloud-

Dienst mit großem Sprachmodell

abfragen. Dann rechnet nicht der

Chip, sondern der Anbieter; du

brauchst einen API-Schlüssel, und die

Daten verlassen dein Netz.

09 Wenn es nicht klappt
SYMPTOM LÖSUNG

„PSRAM: 0 MB“ oder Neustart-Schleife

mit „PSRAM ID read error“

Werkzeuge → PSRAM → OPI PSRAM wählen. Das N16R8 hat Octal-PSRAM –

„QSPI PSRAM“ passt nicht.

Kein Port in der IDE Kabel an die Buchse COM. Anderes USB-C-Kabel probieren. Windows 10/11

erkennt den Wandler meist von selbst, sonst den passenden Treiber des

Chipherstellers WCH von wch-ic.com installieren.

Upload hängt bei „Connecting…“ BOOT gedrückt halten, kurz RST drücken, BOOT loslassen – dann erneut

hochladen.

Monitor bleibt leer 115200 Baud? Kabel an COM? Bei „USB CDC On Boot: Enabled“ geht die

Ausgabe an die andere Buchse.

Nach Beispiel 3 kein Upload mehr über

„USB“

Normal – die Buchse ist jetzt eine Tastatur. Über COM hochladen oder BOOT

+ RST drücken.

GPIO35, 36 oder 37 reagiert nicht Diese Pins sind beim N16R8 vom PSRAM belegt – anderen Pin nehmen.

„Sketch too big“ Partition Scheme mit 3 MB APP wählen (siehe Seite 5).

Bluetooth-Seriell (BluetoothSerial)

fehlt

Der S3 kann nur Bluetooth LE, kein klassisches Bluetooth. Die BLE-Beispiele

im Board-Paket nutzen.

Ständige Neustarts unter Last „Brownout“ – der USB-Port liefert zu wenig Strom. Anderer Port, aktiver Hub

oder kürzeres Kabel.

Passt dazu

INMP441 Mikrofon

Für Sprachbefehle & Audio-

KI

MKR-INMP441-64

GY-521 MPU-6050

Bewegungsdaten für

Gesten-KI

MKR-GY-521-017

WS2812 LED-Ring

Status der KI anzeigen

MKR-WS2812-RING12-67

SHT40 Temp. &

Feuchte

Präzise Sensordaten

MKR-SHT40-76

Sicherheitshinweise. Elektronikmodul für Ausbildungs-,

Entwicklungs- und Experimentierzwecke. Nur mit Kleinspannung

betreiben, niemals an Netzspannung. Der ESP32-S3 wird bei

rechenintensiven Anwendungen spürbar warm – nicht abdecken.

Nicht geeignet für Kinder unter 14 Jahren. Elektroaltgeräte gehören

nicht in den Hausmüll – Rückgabe kostenlos bei kommunalen

Sammelstellen.

Verantwortlicher Wirtschaftsakteur (GPSR): Alexander Prez ·

makeroo · Hinterm Bahnhof 4 · 90513 Zirndorf · support@makeroo.de

Fragen? Schreib uns mit einem Foto deines Aufbaus.

Arduino, ESP32, ESPHome, TensorFlow und Edge Impulse sind

Marken ihrer Inhaber und dienen nur der Kompatibilitätsangabe.

